

# IMP in HOLCF

Tobias Nipkow and Robert Sandner

November 22, 2007

## Contents

<b>1</b>	<b>Syntax of Commands</b>	<b>1</b>
<b>2</b>	<b>Natural Semantics of Commands</b>	<b>2</b>
2.1	Execution of commands . . . . .	2
2.2	Equivalence of statements . . . . .	4
2.3	Execution is deterministic . . . . .	5
<b>3</b>	<b>Denotational Semantics of Commands in HOLCF</b>	<b>7</b>
3.1	Definition . . . . .	7
3.2	Equivalence of Denotational Semantics in HOLCF and Evaluation Semantics in HOL . . . . .	7
<b>4</b>	<b>Correctness of Hoare by Fixpoint Reasoning</b>	<b>8</b>

## 1 Syntax of Commands

`theory Com imports Main begin`

`typeddecl loc`

— an unspecified (arbitrary) type of locations (addresses/names) for variables

`types`

`val = nat` — or anything else, `nat` used in examples

`state = "loc  $\Rightarrow$  val"`

`aexp = "state  $\Rightarrow$  val"`

`bexp = "state  $\Rightarrow$  bool"`

— arithmetic and boolean expressions are not modelled explicitly here,

— they are just functions on states

`datatype`

`com = SKIP`

`/ Assign loc aexp ("_ ::= _" 60)`

`/ Semi com com ("_;_" [60, 60] 10)`

`/ Cond bexp com com ("IF _ THEN _ ELSE _" 60)`

```

      | While bexp com          ("WHILE _ DO _" 60)

syntax (latex)
  SKIP :: com    ("skip")
  Cond  :: "bexp  $\Rightarrow$  com  $\Rightarrow$  com  $\Rightarrow$  com" ("if _ then _ else _" 60)
  While :: "bexp  $\Rightarrow$  com  $\Rightarrow$  com" ("while _ do _" 60)

end

```

## 2 Natural Semantics of Commands

theory Natural imports Com begin

### 2.1 Execution of commands

We write  $\langle c, s \rangle \longrightarrow_c s'$  for *Statement  $c$ , started in state  $s$ , terminates in state  $s'$* . Formally,  $\langle c, s \rangle \longrightarrow_c s'$  is just another form of saying *the tuple  $(c, s, s')$  is part of the relation  $\text{evalc}$* :

```

constdefs
  update :: "('a  $\Rightarrow$  'b)  $\Rightarrow$  'a  $\Rightarrow$  'b  $\Rightarrow$  ('a  $\Rightarrow$  'b)" ("_/_ ::= /_" [900,0,0] 900)
  "update == fun_upd"

```

```

syntax (xsymbols)
  update :: "('a  $\Rightarrow$  'b)  $\Rightarrow$  'a  $\Rightarrow$  'b  $\Rightarrow$  ('a  $\Rightarrow$  'b)" ("_/_  $\mapsto$  /_" [900,0,0] 900)

```

The big-step execution relation  $\text{evalc}$  is defined inductively:

```

inductive
  evalc :: "[com, state, state]  $\Rightarrow$  bool" ("<_,_>/  $\longrightarrow_c$  _" [0,0,60] 60)
where
  Skip:      "<skip, s>  $\longrightarrow_c$  s"
  | Assign:  "<x ::= a, s>  $\longrightarrow_c$  s[x  $\mapsto$  a s]"

  | Semi:    "<c0, s>  $\longrightarrow_c$  s''  $\Longrightarrow$  <c1, s''>  $\longrightarrow_c$  s'  $\Longrightarrow$  <c0; c1, s>  $\longrightarrow_c$  s'"

  | IfTrue:  "b s  $\Longrightarrow$  <c0, s>  $\longrightarrow_c$  s'  $\Longrightarrow$  <if b then c0 else c1, s>  $\longrightarrow_c$  s'"
  | IfFalse: " $\neg$ b s  $\Longrightarrow$  <c1, s>  $\longrightarrow_c$  s'  $\Longrightarrow$  <if b then c0 else c1, s>  $\longrightarrow_c$  s'"

  | WhileFalse: " $\neg$ b s  $\Longrightarrow$  <while b do c, s>  $\longrightarrow_c$  s"
  | WhileTrue:  "b s  $\Longrightarrow$  <c, s>  $\longrightarrow_c$  s''  $\Longrightarrow$  <while b do c, s''>  $\longrightarrow_c$  s'
                 $\Longrightarrow$  <while b do c, s>  $\longrightarrow_c$  s'"

```

lemmas  $\text{evalc.intros}$  [intro] — use those rules in automatic proofs

The induction principle induced by this definition looks like this:

$$\llbracket \langle x1, x2 \rangle \longrightarrow_c x3; \bigwedge s. P \text{ skip } s \text{ s}; \bigwedge x \text{ a } s. P (x ::= a) s (s[x \mapsto a s]); \bigwedge c0 \text{ s } s'' \text{ c1 } s'.$$

$$\begin{aligned}
& \llbracket \langle c0, s \rangle \longrightarrow_c s''; P \ c0 \ s \ s''; \langle c1, s'' \rangle \longrightarrow_c s'; P \ c1 \ s'' \ s' \rrbracket \\
& \implies P \ (c0; c1) \ s \ s'; \\
& \bigwedge b \ s \ c0 \ s' \ c1. \ \llbracket b \ s; \langle c0, s \rangle \longrightarrow_c s'; P \ c0 \ s \ s' \rrbracket \implies P \ (\text{if } b \text{ then } c0 \text{ else } c1) \ s \ s'; \\
& \bigwedge b \ s \ c1 \ s' \ c0. \ \llbracket \neg b \ s; \langle c1, s \rangle \longrightarrow_c s'; P \ c1 \ s \ s' \rrbracket \implies P \ (\text{if } b \text{ then } c0 \text{ else } c1) \ s \\
& s'; \\
& \bigwedge b \ s \ c. \ \neg b \ s \implies P \ (\text{while } b \text{ do } c) \ s \ s; \\
& \bigwedge b \ s \ c \ s'' \ s'. \\
& \quad \llbracket b \ s; \langle c, s \rangle \longrightarrow_c s''; P \ c \ s \ s''; \langle \text{while } b \text{ do } c, s'' \rangle \longrightarrow_c s'; \\
& \quad P \ (\text{while } b \text{ do } c) \ s'' \ s' \rrbracket \\
& \implies P \ (\text{while } b \text{ do } c) \ s \ s' \\
& \implies P \ x1 \ x2 \ x3
\end{aligned}$$

( $\bigwedge$  and  $\implies$  are Isabelle's meta symbols for  $\forall$  and  $\longrightarrow$ )

The rules of `evalc` are syntax directed, i.e. for each syntactic category there is always only one rule applicable. That means we can use the rules in both directions. The proofs for this are all the same: one direction is trivial, the other one is shown by using the `evalc` rules backwards:

**lemma skip:**

```
"⟨skip, s⟩ ⟶c s' = (s' = s)"
by (rule, erule evalc.cases) auto
```

**lemma assign:**

```
"⟨x ::= a, s⟩ ⟶c s' = (s' = s[x ↦ a])"
by (rule, erule evalc.cases) auto
```

**lemma semi:**

```
"⟨c0; c1, s⟩ ⟶c s' = (∃ s''. ⟨c0, s⟩ ⟶c s'' ∧ ⟨c1, s''⟩ ⟶c s')"
by (rule, erule evalc.cases) auto
```

**lemma ifTrue:**

```
"b s ⟹ ⟨if b then c0 else c1, s⟩ ⟶c s' = ⟨c0, s⟩ ⟶c s'"
by (rule, erule evalc.cases) auto
```

**lemma ifFalse:**

```
"¬ b s ⟹ ⟨if b then c0 else c1, s⟩ ⟶c s' = ⟨c1, s⟩ ⟶c s'"
by (rule, erule evalc.cases) auto
```

**lemma whileFalse:**

```
"¬ b s ⟹ ⟨while b do c, s⟩ ⟶c s' = (s' = s)"
by (rule, erule evalc.cases) auto
```

**lemma whileTrue:**

```
"b s ⟹
⟨while b do c, s⟩ ⟶c s' =
(∃ s''. ⟨c, s⟩ ⟶c s'' ∧ ⟨while b do c, s''⟩ ⟶c s'"
by (rule, erule evalc.cases) auto
```

Again, Isabelle may use these rules in automatic proofs:

```
lemmas evalc_cases [simp] = skip assign ifTrue ifFalse whileFalse semi whileTrue
```

## 2.2 Equivalence of statements

We call two statements  $c$  and  $c'$  equivalent wrt. the big-step semantics when  $c$  started in  $s$  terminates in  $s'$  iff  $c'$  started in the same  $s$  also terminates in the same  $s'$ . Formally:

**constdefs**

```
equiv_c :: "com  $\Rightarrow$  com  $\Rightarrow$  bool" ("_  $\sim$  _")
" $c \sim c' \equiv \forall s\ s'. \langle c, s \rangle \longrightarrow_c s' = \langle c', s \rangle \longrightarrow_c s'$ "
```

Proof rules telling Isabelle to unfold the definition if there is something to be proved about equivalent statements:

**lemma equivI [intro!]:**

```
"( $\bigwedge s\ s'. \langle c, s \rangle \longrightarrow_c s' = \langle c', s \rangle \longrightarrow_c s'$ )  $\implies c \sim c'$ "
by (unfold equiv_c_def) blast
```

**lemma equivD1:**

```
" $c \sim c' \implies \langle c, s \rangle \longrightarrow_c s' \implies \langle c', s \rangle \longrightarrow_c s'$ "
by (unfold equiv_c_def) blast
```

**lemma equivD2:**

```
" $c \sim c' \implies \langle c', s \rangle \longrightarrow_c s' \implies \langle c, s \rangle \longrightarrow_c s'$ "
by (unfold equiv_c_def) blast
```

As an example, we show that loop unfolding is an equivalence transformation on programs:

**lemma unfold\_while:**

```
"(while b do c)  $\sim$  (if b then c; while b do c else skip)" (is "?w  $\sim$  ?if")
```

**proof -**

— to show the equivalence, we look at the derivation tree for

— each side and from that construct a derivation tree for the other side

```
{ fix s s' assume w: " $\langle ?w, s \rangle \longrightarrow_c s'$ "
```

— as a first thing we note that, if  $b$  is *False* in state  $s$ ,

— then both statements do nothing:

```
hence " $\neg b\ s \implies s = s'$ " by simp
```

```
hence " $\neg b\ s \implies \langle ?if, s \rangle \longrightarrow_c s'$ " by simp
```

**moreover**

— on the other hand, if  $b$  is *True* in state  $s$ ,

— then only the *WhileTrue* rule can have been used to derive  $\langle ?w, s \rangle \longrightarrow_c s'$

```
{ assume b: "b s"
```

**with w obtain s'' where**

```
" $\langle c, s \rangle \longrightarrow_c s''$ " and " $\langle ?w, s'' \rangle \longrightarrow_c s'$ " by (cases set: evalc) auto
```

— now we can build a derivation tree for the if

— first, the body of the True-branch:

```
hence " $\langle c; ?w, s \rangle \longrightarrow_c s'$ " by (rule Semi)
```

— then the whole if

```
with b have " $\langle ?if, s \rangle \longrightarrow_c s'$ " by (rule IfTrue)
```

```
}
```

```

ultimately
  — both cases together give us what we want:
  have "<?if, s> →c s'" by blast
}
moreover
  — now the other direction:
  { fix s s' assume "if": "<?if, s> →c s'"
    — again, if b is False in state s, then the False-branch
    — of the if is executed, and both statements do nothing:
    hence "¬b s ⇒ s = s'" by simp
    hence "¬b s ⇒ <?w, s> →c s'" by simp
    moreover
      — on the other hand, if b is True in state s,
      — then this time only the IfTrue rule can have be used
      { assume b: "b s"
        with "if" have "<c; ?w, s> →c s'" by (cases set: evalc) auto
        — and for this, only the Semi-rule is applicable:
        then obtain s'' where
          "<c, s> →c s'" and "<?w, s''> →c s'" by (cases set: evalc) auto
        — with this information, we can build a derivation tree for the while
        with b
        have "<?w, s> →c s'" by (rule WhileTrue)
      }
    ultimately
      — both cases together again give us what we want:
      have "<?w, s> →c s'" by blast
  }
ultimately
  show ?thesis by blast
qed

```

## 2.3 Execution is deterministic

The following proof presents all the details:

```

theorem com_det:
  assumes "<c, sc t" and "<c, sc u"
  shows "u = t"
  using prems
proof (induct arbitrary: u set: evalc)
  fix s u assume "<skip, sc u"
  thus "u = s" by simp
next
  fix a s x u assume "<x ::= a, sc u"
  thus "u = s[x ↦ a s]" by simp
next
  fix c0 c1 s s1 s2 u
  assume IH0: "∧u. <c0, sc u ⇒ u = s2"
  assume IH1: "∧u. <c1, s2c u ⇒ u = s1"

```

```

assume " $\langle c0; c1, s \rangle \longrightarrow_c u$ "
then obtain  $s'$  where
   $c0$ : " $\langle c0, s \rangle \longrightarrow_c s'$ " and
   $c1$ : " $\langle c1, s' \rangle \longrightarrow_c u$ "
  by auto

from  $c0$  IH0 have " $s' = s2$ " by blast
with  $c1$  IH1 show " $u = s1$ " by blast
next
fix  $b$   $c0$   $c1$   $s$   $s1$   $u$ 
assume IH: " $\bigwedge u. \langle c0, s \rangle \longrightarrow_c u \implies u = s1$ "

assume " $b$   $s$ " and " $\langle \text{if } b \text{ then } c0 \text{ else } c1, s \rangle \longrightarrow_c u$ "
hence " $\langle c0, s \rangle \longrightarrow_c u$ " by simp
with IH show " $u = s1$ " by blast
next
fix  $b$   $c0$   $c1$   $s$   $s1$   $u$ 
assume IH: " $\bigwedge u. \langle c1, s \rangle \longrightarrow_c u \implies u = s1$ "

assume " $\neg b$   $s$ " and " $\langle \text{if } b \text{ then } c0 \text{ else } c1, s \rangle \longrightarrow_c u$ "
hence " $\langle c1, s \rangle \longrightarrow_c u$ " by simp
with IH show " $u = s1$ " by blast
next
fix  $b$   $c$   $s$   $u$ 
assume " $\neg b$   $s$ " and " $\langle \text{while } b \text{ do } c, s \rangle \longrightarrow_c u$ "
thus " $u = s$ " by simp
next
fix  $b$   $c$   $s$   $s1$   $s2$   $u$ 
assume "IHc": " $\bigwedge u. \langle c, s \rangle \longrightarrow_c u \implies u = s2$ "
assume "IHw": " $\bigwedge u. \langle \text{while } b \text{ do } c, s2 \rangle \longrightarrow_c u \implies u = s1$ "

assume " $b$   $s$ " and " $\langle \text{while } b \text{ do } c, s \rangle \longrightarrow_c u$ "
then obtain  $s'$  where
   $c$ : " $\langle c, s \rangle \longrightarrow_c s'$ " and
   $w$ : " $\langle \text{while } b \text{ do } c, s' \rangle \longrightarrow_c u$ "
  by auto

from  $c$  "IHc" have " $s' = s2$ " by blast
with  $w$  "IHw" show " $u = s1$ " by blast
qed

```

This is the proof as you might present it in a lecture. The remaining cases are simple enough to be proved automatically:

```

theorem
  assumes " $\langle c, s \rangle \longrightarrow_c t$ " and " $\langle c, s \rangle \longrightarrow_c u$ "
  shows " $u = t$ "
  using prems
proof (induct arbitrary:  $u$ )
  — the simple skip case for demonstration:

```

```

fix s u assume "<skip,s> →c u"
thus "u = s" by simp
next
— and the only really interesting case, while:
fix b c s s1 s2 u
assume "IHc": "∧u. <c,s> →c u ⇒ u = s2"
assume "IHw": "∧u. <while b do c,s2> →c u ⇒ u = s1"

assume "b s" and "<while b do c,s> →c u"
then obtain s' where
  c: "<c,s> →c s'" and
  w: "<while b do c,s'> →c u"
  by auto

from c "IHc" have "s' = s2" by blast
with w "IHw" show "u = s1" by blast
qed (best dest: evalc_cases [THEN iffD1])+ — prove the rest automatically

end

```

### 3 Denotational Semantics of Commands in HOLCF

theory Denotational imports HOLCF Natural begin

#### 3.1 Definition

definition

```

dlift :: "('a::type) discr -> 'b::pcpo) => ('a lift -> 'b)" where
"dlift f = (LAM x. case x of UU => UU | Def y => f.(Discr y))"

```

```

consts D :: "com => state discr -> state lift"

```

primrec

```

"D(skip) = (LAM s. Def(undiscr s))"
"D(X ::= a) = (LAM s. Def((undiscr s)[X ↦ a(undiscr s)]))"
"D(c0 ; c1) = (dlift(D c1) oo (D c0))"
"D(if b then c1 else c2) =
  (LAM s. if b (undiscr s) then (D c1)·s else (D c2)·s)"
"D(while b do c) =
  fix·(LAM w s. if b (undiscr s) then (dlift w)·((D c)·s)
    else Def(undiscr s))"

```

#### 3.2 Equivalence of Denotational Semantics in HOLCF and Evaluation Semantics in HOL

```

lemma dlift_Def [simp]: "dlift f.(Def x) = f.(Discr x)"
  by (simp add: dlift_def)

```

```

lemma cont_dlift [iff]: "cont (%f. dlift f)"
  by (simp add: dlift_def)

lemma dlift_is_Def [simp]:
  "(dlift f.l = Def y) = ( $\exists x. l = \text{Def } x \wedge f.(\text{Discr } x) = \text{Def } y$ )"
  by (simp add: dlift_def split: lift.split)

lemma eval_implies_D: " $\langle c, s \rangle \longrightarrow_c t \implies D\ c.(\text{Discr } s) = (\text{Def } t)$ "
  apply (induct set: evalc)
  apply simp_all
  apply (subst fix_eq)
  apply simp
  apply (subst fix_eq)
  apply simp
  done

lemma D_implies_eval: "!s t. D c.(Discr s) = (Def t) -->  $\langle c, s \rangle \longrightarrow_c t$ "
  apply (induct c)
  apply simp
  apply simp
  apply force
  apply (simp (no_asm))
  apply force
  apply (simp (no_asm))
  apply (rule fix_ind)
  apply (fast intro!: adm_lemmas adm_chfindom ax_flat)
  apply (simp (no_asm))
  apply (simp (no_asm))
  apply safe
  apply fast
  done

theorem D_is_eval: "(D c.(Discr s) = (Def t)) = ( $\langle c, s \rangle \longrightarrow_c t$ )"
  by (fast elim!: D_implies_eval [rule_format] eval_implies_D)

end

```

## 4 Correctness of Hoare by Fixpoint Reasoning

theory HoareEx imports Denotational begin

An example from the HOLCF paper by Müller, Nipkow, Oheimb, Slotosch [1]. It demonstrates fixpoint reasoning by showing the correctness of the Hoare rule for while-loops.

types assn = "state => bool"

definition

hoare\_valid :: "[assn, com, assn] => bool" ("|= {(1\_)} / ( \_ ) / {(1\_)}" 50) where  
 "|= {A} c {B} = ( $\forall s\ t. A\ s \wedge D\ c\ \$(\text{Discr } s) = \text{Def } t \longrightarrow B\ t$ )"



```

lemma WHILE_rule_sound:
  " $\models \{A\} c \{A\} \implies \models \{A\} \text{ while } b \text{ do } c \{\lambda s. A s \wedge \neg b s\}$ "
  apply (unfold hoare_valid_def)
  apply (simp (no_asm))
  apply (rule fix_ind)
  apply (simp (no_asm)) — simplifier with enhanced adm-tactic
  apply (simp (no_asm))
  apply (simp (no_asm))
  apply blast
done

end

```

## References

- [1] O. Müller, T. Nipkow, D. v. Oheimb, and O. Slotosch. HOLCF = HOL + LCF. *J. Functional Programming*, 9:191–223, 1999.